Miscellaneous problems

Q1.1

```
> with(plots):
> display(
   plot([cos(t)/2-cos(2*t)/4,sin(t)/2-sin(2*t)/4,t=0..2*Pi]),
   plot([cos(t)/4-1,sin(t)/4,t=0..2*Pi]),
   plot([-0.1225+0.0945*cos(t), 0.7449+0.0945*sin(t),t=0..2*Pi]),
   plot([-0.1225+0.0945*cos(t),-0.7449+0.0945*sin(t),t=0..2*Pi]),
   scaling=constrained,axes=none
  );
```



Here is a picture of the full Mandelbrot set:



Q1.2:

```
> with(plots):
```

Here is the picture for $n = 6$:

```
> implicitplot(abs(x)^6+abs(y)^6=6^(6/2),x=-3..3,y=-3..3,grid=[200,
  200]);
```



Here are $n = 2$ and $n = 6$ together:

```
> display(
   implicitplot(abs(x)^2+abs(y)^2=2^(2/2),x=-3..3,y=-3..3,grid=
```

```
[200,200]),
 implicitplot(abs(x)^6+abs(y)^6=6^(6/2),x=-3..3,y=-3..3,grid=
[200,200])
);
```



Here is the full picture:

```
> display(
   seq(
    implicitplot(abs(x)^n+abs(y)^n=n^(n/2),
                 x=-3..3,y=-3..3,grid=[200,200]),
    n=1..9)
  );
```



Q1.3:

```
> listplot([seq(50^n/n!,n=0..100)],style=point);
```



Q1.4:

```
> y := exp(-1/x);
```

$$y := e^{-\frac{1}{x}} \qquad (1)$$

We can work out the first few derivatives as follows:

```
> diff(y,x);
```

$$\frac{e^{-\frac{1}{x}}}{x^2} \quad (2)$$

```
> simplify(diff(y,x$2));
```

$$-\frac{e^{-\frac{1}{x}}\,(2\,x-1)}{x^4} \quad (3)$$

```
> simplify(diff(y,x$3));
```

$$\frac{e^{-\frac{1}{x}}\,\left(6\,x^2-6\,x+1\right)}{x^6} \quad (4)$$

```
> simplify(diff(y,x$4));
```

$$-\frac{e^{-\frac{1}{x}}\,\left(24\,x^3-36\,x^2+12\,x-1\right)}{x^8} \quad (5)$$

```
> simplify(diff(y,x$5));
```

$$\frac{e^{-\frac{1}{x}}\,\left(120\,x^4-240\,x^3+120\,x^2-20\,x+1\right)}{x^{10}} \quad (6)$$

We see that $\frac{\partial^k}{\partial x^k}\,y$ is always of the form $\frac{e^{-\frac{1}{x}}\,p_k(x)}{x^{2\,k}}$ for some polynomial $p_k(x)$. Explicitly, the first ten of these
polynomials are as follows:

```
> for k from 1 to 10 do p[k] := expand(x^(2*k)*diff(y,x$k)/y); od;
```

$$p_1 := 1 \quad (7)$$

$$p_2 := -2\,x+1$$

$$p_3 := 6\,x^2-6\,x+1$$

$$p_4 := -24\,x^3+36\,x^2-12\,x+1$$

$$p_5 := 120\,x^4-240\,x^3+120\,x^2-20\,x+1$$

$$p_6 := -720\,x^5+1800\,x^4-1200\,x^3+300\,x^2-30\,x+1$$

$$p_7 := 5040\,x^6-15120\,x^5+12600\,x^4-4200\,x^3+630\,x^2-42\,x+1$$

$$p_8 := -40320\,x^7+141120\,x^6-141120\,x^5+58800\,x^4-11760\,x^3+1176\,x^2-56\,x+1$$

$$p_9 := 362880\,x^8-1451520\,x^7+1693440\,x^6-846720\,x^5+211680\,x^4-28224\,x^3+2016\,x^2 \\ -72\,x+1$$

$$p_{10} := -3628800\,x^9 + 16329600\,x^8 - 21772800\,x^7 + 12700800\,x^6 - 3810240\,x^5 + 635040\,x^4 - 60480\,x^3 + 3240\,x^2 - 90\,x + 1$$

We see from this that $p_k(x)$ always has degree $k-1$. The constant term is always $p_k(0)=1$. You should recognise
the numbers 2, 6, 24, 120, 720 as factorials, so the highest term in $p_k(x)$ is $(-1)^k k!\, x^{k-1}$.

We can plot the derivatives up to $k=4$ as follows. We have divided each function by its approximate maximum value so that we can see them clearly in the same picture. We find that the higher derivatives
oscillate wildly for moderately small values of x, but then flatten out for very small values of x, and also
for reasonably large values of x..

```
> plot([
  y/0.6,
  diff(y,x)/0.55,
  diff(y,x,x)/2.5,
  diff(y,x,x,x)/30,
  diff(y,x,x,x,x)/600],
  x=0..2);
```



This plot of the sixth derivative reinforces the same message.

```
> plot(diff(y,x$6),x=0..3,-100..100);
```



We can watch the curves flattening out near the origin as follows:

```
> plot([
  y,
  diff(y,x),
  diff(y,x,x),
  diff(y,x,x,x),
  diff(y,x,x,x,x)],
  x=0..0.1,0..0.002);
```



Solitons

```
> with(plots):
```

We can enter the basic definitions as follows.

```
> q := sqrt(2);
  p := log(3 + 2*q);
  r := x-4*t;
  s := q*(x - 8*t);
  T := 32 * cosh(2*r-p) + 16 * cosh(2*s-p) + 16;
  B := 4*(1+q) * cosh(r)*cosh(s) + (4*q-8)*exp(r+s);
  phi[0] := 2*cosh(r)^(-2);
  phi[1] := 4*cosh(s)^(-2);
  phi[2] := 2*cosh(r-p)^(-2);
  phi[3] := 4*cosh(s-p)^(-2);
  phi[4] := T/B^2;
```

$$q := \sqrt{2} \tag{8}$$

$$p := \ln\left(3 + 2\sqrt{2}\right)$$

$$r := x - 4\,t$$

$$s := \sqrt{2}\,\left(x - 8\,t\right)$$

$$T := 32\cosh\left(-2\,x + 8\,t + \ln\left(3 + 2\sqrt{2}\right)\right) + 16\cosh\left(-2\sqrt{2}\,\left(x - 8\,t\right) + \ln\left(3 + 2\sqrt{2}\right)\right) + 16$$

$$B := 4\,\left(1 + \sqrt{2}\right)\cosh(-x + 4\,t)\cosh\left(\sqrt{2}\,\left(x - 8\,t\right)\right) + \left(4\sqrt{2} - 8\right)\mathrm{e}^{x - 4\,t + \sqrt{2}\,(x - 8\,t)}$$

$$\phi_0 := \frac{2}{\cosh(-x + 4\,t)^2}$$

$$\phi_1 := \frac{4}{\cosh\left(\sqrt{2}\,\left(x - 8\,t\right)\right)^2}$$

$$\phi_2 := \frac{2}{\cosh\left(-x + 4\,t + \ln\left(3 + 2\sqrt{2}\right)\right)^2}$$

$$\phi_3 := \frac{4}{\cosh\left(-\sqrt{2}\,\left(x - 8\,t\right) + \ln\left(3 + 2\sqrt{2}\right)\right)^2}$$

$$\phi_4 := \left(32\cosh\left(-2\,x + 8\,t + \ln\left(3 + 2\sqrt{2}\right)\right) + 16\cosh\left(-2\sqrt{2}\,\left(x - 8\,t\right) + \ln\left(3 + 2\sqrt{2}\right)\right) + 16\right) \Big/ \left(4\,\left(1 + \sqrt{2}\right)\cosh(-x + 4\,t)\cosh\left(\sqrt{2}\,\left(x - 8\,t\right)\right) + \left(4\sqrt{2}\right.\right.$$

$$-8)\, e^{x-4t+\sqrt{2}\,(x-8t)}\Big)^2$$

We now check that the Korteweg-de Vries equation is satisfied. The tidiest way is to introduce the KdV operator as follows:

```
> K := (u) -> diff(u,t) + diff(u,x,x,x) + 6 * u * diff(u,x);
```

$$K := u \rightarrow \frac{\partial}{\partial t}\,u + \frac{\partial^3}{\partial x^3}\,u + 6\,u\left(\frac{\partial}{\partial x}\,u\right) \tag{9}$$

We now apply K to the functions ϕ_i:

```
> simplify(K(phi[0]));
  simplify(K(phi[1]));
  simplify(K(phi[2]));
  simplify(K(phi[3]));
```

$$0 \tag{10}$$
$$0$$
$$0$$
$$0$$

If we just apply K directly to ϕ_4 then we get several pages of dense output, because Maple is not very good at simplifying
hyperbolic functions. As explained in the problem sheet, we need to convert everything to exponential form first.

```
> convert(phi[4],exp);
```

$$\left(16\,e^{-2x+8t+\ln(3+2\sqrt{2})} + 16\,e^{2x-8t-\ln(3+2\sqrt{2})} + 8\,e^{-2\sqrt{2}\,(x-8t)+\ln(3+2\sqrt{2})} + 8\,e^{2\sqrt{2}\,(x-8t)-\ln(3+2\sqrt{2})} + 16\right) \Big/ \left(4\,(1+\sqrt{2})\left(\frac{1}{2}\,e^{-x+4t} + \frac{1}{2}\,e^{x-4t}\right)\left(\frac{1}{2}\,e^{\sqrt{2}\,(x-8t)} + \frac{1}{2}\,e^{-\sqrt{2}\,(x-8t)}\right) + (4\sqrt{2}-8)\,e^{x-4t+\sqrt{2}\,(x-8t)}\right)^2 \tag{11}$$

```
> simplify(K(convert(phi[4],exp)));
```

$$0 \tag{12}$$

We can plot all the functions ϕ_i together as follows. The bottom one is ϕ_0 and the top one is ϕ_4.

```
> animate(
  plot,
  [[phi[0],phi[1]+5,phi[2]+10,phi[3]+15,phi[4]+20],x=-30..30],
  t=-3..3,
```

```
frames=50,
scaling=constrained,
axes=none);
```



We see that ϕ_0 is essentially the same as ϕ_2 but delayed slightly; this is already easy to see from the formulae.
Similarly, ϕ_1 is a slightly delayed copy of ϕ_3. When t is negative, the small hump in ϕ_4 follows ϕ_2 and the

large hump in ϕ_4 follows ϕ_1, so ϕ_4 is approximately $\phi_1 + \phi_2$. Near $t = 0$ the two humps interact, the large one
jumps forward a little to follow ϕ_3, and the small hump drops back a little to follow ϕ_0. Thus, when t is positive
we see that ϕ_4 is approximately $\phi_0 + \phi_3$.

We can see all this again in the following animation. The middle graph is ϕ_4. The bottom graph is $\phi_4 - \phi_1 - \phi_2$;
we find that this is very small when t is negative. The top graph is $\phi_4 - \phi_0 - \phi_3$; we find that this is very small when
t is positive.

```
> animate(
  plot,
  [[phi[4]-phi[1]-phi[2]-10,phi[4],phi[4]-phi[0]-phi[3]+10],x=-30.
  .30],
  t=-3..3,
  frames=50,
  scaling=constrained,
  axes=none);
```

t = -3.



We now calculate the momenta of $\phi_0 .. \phi_3$:

```
> M[0] := int(phi[0],x=-infinity..infinity);
  M[1] := int(phi[1],x=-infinity..infinity);
  M[2] := int(phi[2],x=-infinity..infinity);
  M[3] := int(phi[3],x=-infinity..infinity);
```

$$M_0 := 4 \tag{13}$$

$$M_1 := 4\sqrt{2}$$

$$M_2 := 4$$

$$M_3 := 4\sqrt{2}$$

We can do this more efficiently as follows:

```
> for i from 0 to 3 do
   M[i] := int(phi[i],x=-infinity..infinity);
  od;
```

$$M_0 := 4 \quad \textbf{(14)}$$

$$M_1 := 4\sqrt{2}$$

$$M_2 := 4$$

$$M_3 := 4\sqrt{2}$$

For M_4 we need a more elaborate method, as described in the problem sheet:

```
> M[4] := int(subs(t=0,phi[4]),x=-infinity..infinity,numeric,
  method=_Gquad);
```

$$M_4 := 9.656854250 \quad \textbf{(15)}$$

We find that $M_4 = M_1 + M_2$, apart from a tiny error that would go away if we computed the integral more accurately.

```
> evalf(M[4] - M[1] - M[2]);
```

$$2.\ 10^{-9} \quad \textbf{(16)}$$

We now repeat this for the energy:

```
> for i from 0 to 3 do
   E[i] := int(phi[i]^2,x=-infinity..infinity);
  od;
```

$$E_0 := \frac{16}{3} \quad \textbf{(17)}$$

$$E_1 := \frac{32\sqrt{2}}{3}$$

$$E_2 := \frac{16}{3}$$

$$E_3 := \frac{32\sqrt{2}}{3}$$

```
> E[4] := int(subs(t=0,phi[4]^2),x=-infinity..infinity,numeric,
  method=_Gquad);
```

$$E_4 := 20.41827800 \quad \textbf{(18)}$$

```
> evalf(E[4] - E[1] - E[2]);
```

$$0. \quad \textbf{(19)}$$